
 ALGORITHMS AND DATA STRUCTURES

8  Search Table I

8.1  Access time.

Access time is the time that has passed from the request for item with the searched key to the time when the item is usable or when the predicate "search" is set to true or false. There are several access times:

  minimal search time
  maximal search time (the worst case time)
  average search time (the sum of search times of all keys divided by their number) 
 
  The access time for successful and not successful search may differ.

8.2  Basic structure of the search algorithm.

Search algorithm, which searches the item with given key in the set of items has the form of cycle:

       found:=false;
       while <not found> and <set of items is not exhausted> do
          <examine the next item and set found to true if key found >
       Search:=found;

8.3 Classification of search algorithms

According to the access to individual items of the set of items the algorithms can be divided to:

 1. sequential algorithms -  the next item may be examined only after its predecessor
 2. algorithms with the random access - all items are accessible immediately

According to processing of the set of items (and according to the type of the structure implementing the set of items) algorithms can be classified as:
   1 Sequential searching in not ordered sequential structure
   2.Sequential searching in array with the sentinel (guard)
   3.Sequential searching in ordered sequential structure (Sequential structures may be:array, list, file)
   4.Sequential searching in ordered array with the sentinel (guard)
   5.Sequential searching in structure ordered according to the probability of being searched. (Searching in probability adaptive  array)
   6.Binary search in the ordered array. (Normal binary search, Dijkstra binary search for multiple keys)
   7. Fibonacci search
   8. Binary search trees (BST)  (BST with the back pointers, AVL trees)
   9. Hash-tables


¨

8.4  Sequential searching in not ordered sequential structure

Let us take the type

type
   TPtr=^TItem; (* for table implemented by the list *)
   TItem=record
           Key:TKey;  (* operation equal is defined upon the TKey *)
           Data:TData;
           Next:TPtr;  (* only if item is a member of list *)
         end;
    TArray=array[1..Max] of TItem;
    TFile=file of TItem;
    TList=TPtr;

   Search cycle for the array:

   function Search(Table:TArray; K:TKey):Boolean;
(* algorithm returns true/false in the name of the function *)
   var
     i:integer;
     Found:Boolean;
   begin
     i:=1;
     Found:=false;
     while not Found and (i<=Max) do begin
        Found:=Table[i].Key=K;
        i:=i+1;
     end;
     Search:=Found;
   end;

Note:    the cycle

     while (Table[i].Key<>K and (i<=Max)  do...
     is not correct. 

If i>Max the Table[i] is out of range!!!

     The cycle: 
     while (i<=Max) and (Table[i].Key<>K do...

     is correct even for "short-cut evaluation of Boolean expression".  We avoid this style unless we have some special reason for it!
   

    Search cycle for the file has the form:



    function Search(var F:TFile; K:TKey):Boolean;
    var
      Found:Boolean;
      Tmp:TItem;
    begin
      Found:=false;
      reset(F);
      while not Found and not eof(F) do begin
        Read(F,Tmp);
        Found:=Tmp.Key=K
      end;
      Search:=Found
    end;


    Search cycle for the list has the form:

    function Search(T:Tptr; K:TKey):Boolean;
    var
      Found:Boolean;
    begin
      Found:=false;
      while not Found and (T<>nil) do begin
        Found:=T^.Key=K;
        T:=T^.Next
      end;
      Search:=Found;
    end;

   Similarly the cycle:
     while (T^.Key <> K) and (T<>nil) do T:=T^.Next;

   is not correct as for T=nil the T^.Key is not defined;

   Reversed Boolean expression is possible only with short-cut evaluation of Boolean expression:

     while (T<>nil) and (T^.Key <> K) do T:=T^.Next;

  We avoid this style unless we have some special reason for it!
   
====================================================================
8.4.1  Exercise:

Write the search cycle for search table implemented by an array or by a list, so that the output parameter "where" gives an access (index or pointer) to the searched item. If the search is false the value of Where is not defined.


procedure Search(T:TArray; Key:TKey; var Found:Boolean,
                 var Where:integer); { assuming the type of data is integer }
(* procedure heading for table implemented by an array; Value of Where
will not be defined for Found=false *)

procedure Search(T:TPtr; Key:TKey; var Found:Boolean; var Where:TPtr);
(* procedure heading for table  implemented by a list. Value of Where will not be defined for Found=false  *)

variants:

procedure Search(T:TArray; Key:TKey; var Where:integer);
(* if Where=0 then Search = false
                  else index Where points to the found item *)

procedure Search(T:TPtr; Key:TKey; var Where:TPtr);
(*  if Where = nil then Search=false 
                    else pointer Where points to the found item *)
======================================================================

8.4.2  Table implemented by not ordered array

There is N items in the array (N<=Max)

Search is mentioned above;

Operation Insert

procedure Insert(var T:TArr; var N:Integer; K:TKey; El:TData; var Full:Boolean);
(* N is the number of items contained on the table. N<=Max *)
var
  Found:Boolean;
  Where:integer;
begin
  Full:=false;
  Search(T,N,K,Found,Where);
  (* N is the number of items in the array *)
  if Found
  then T[Where].Data:=El
  else begin
    if N=Max
    then Full:=true
    else begin
      N:=N+1;
      T[N].Key:=K;
      T[N].Data:=El
    end
  end;
end;



Operation Delete


procedure Delete(var T:TArr; var N:integer; K:Tkey);
var
   Found:Boolean;
   Where:integer;
begin
  Search(T,N,K,Found,Where);
  if Found 
  then begin
    T[Where] := T[N];   
    N:=N-1             
  end
end; 

======================================================================
   
8.4.3  Search with the sentinel (guard) or "quick linear search"

procedure Search(T:TArr;N:integer;K:TKey;var Found:Boolean;var 
                 Where:integer);
var
begin
  T[N+1].Key:=K;   (* setting the sentinel (or guard) *)
  Where:=1;
  while T[Where].Key<>K do Where:=Where+1;
  Found:=Where<>(N+1)
end;
   
Note: The Key is stored at the position N+1. This position is called "a 
sentinel" or "a guard". The cycle finished with success in search, but 
the real Found is true only if the cycle ends on the index different 
from N+1! The algorithm is faster than previous as it doesn't evaluate 
two-part Boolean expression every pass.

=====================================================================

  8.4.4 Sequential searching in ordered sequential structure
     (Sequential structures may be: array,list, file)

The sequence of items in the sequential structure may be ordered by the value of the key. It is possible if there is relation operation defined upon the type TKey!

The only advantage of the ordered sequence is faster not successful Search !!!

procedure Search (T:TArr;N:integer;K:TKey;var Found:Boolean; var Where:integer);
var
  i:integer;
  Stop:Boolean; (* Auxiliaryn variable *)
begin
  Where:=1;
  Stop:=false;
  while not Stop and (Where<=N) do begin
    Stop:=K<=T[Where].Key; (* stop if processed item has the key equal 
                              or less the searched key !!! *)
  end;
  Found:=T[Where].Key=K;
  (* if Found=False, then Where points to the place to which the new 
     item may  be placed! *)

   (* for K=7, N=6,
   T =  1,3,5,8,10,12
                    ^
            Where
       Found=false,  Where=4,  as the item 8 has index 4*)
end;

======================================================================
Exercise:

1.  Write the similar procedure for the list implemented table.
2.  Write the similar procedure for the array implemented table with the  sentinel (guard).

8.4.5 Operation Insert in ordered array

Insertion of new item should preserve ordering of all items. The free space for the new item should be made by shift of the massive of items to the right.


procedure Insert (var T:TArr;var N:integer;K:TKey;El:TData; 
                  var  Full:Boolean);
var
  Found:Boolean;
  Where,i:integer;

begin
  Full:=false;
  Search(T,N,K,Found,Where);
  if Found
  then begin
    T[Where].Data:=El;  (* Actualization of Data *)
  else begin
    if N=Max
    then Full:=true
    else begin
      for i:=N-1 downto Where do T[i+1]:=T[i];  (* Shift of the massive of
                                               items - making free space 
                                               for new item *)
      T[Where].Key:=K;        (* Insertion of the Key *)
      T[Where].Data:=El;      (* Insertion of the Data *)
      N:=N+1
  end;
end;

procedure Delete(var T:TArr; var N:integer; K:TKey);
var
  Found:Boolean;
  Where,i:integer;
begin
  Search(T,N,K,Found,Where);
  if Found
  then begin 
    for i:=Where to N-1 do begin T[i]:=T[i+1]
    (* The cycle shifts the massive of items one possiton to the left 
       and thus rewrites the deleted item *)
    N:=N-1;
  end;
end;

======================================================================


8.4.5  Sequential searching in structure ordered according to the probability of being searched. (Serching in probability adaptive array)

If the probability of searching all the keys is not equal, the table may 
be ordered according to the probability of being searched. 

procedure Search(var T:TArr; N:integer; K:TKey; var Found:Boolean; 
                 var Where:integer);
begin
  Where:=1;
  Found:=false;
  while not Found and Where<=N do begin
     If T[Where].Key=K
     then Found:=true
     else Where:=Where+1;
  end;
  if Found and (Where<>1)
  then T[Where]:=:T[Where-1]  (* operation  ":=:" is our shorthand 
                                 notation for swapping of the two elements, 
                                 i.e. :   A:=:B is the same like   
                                 P:=A; A:=B; B:=P   *)
      
end;

Note: If Search finds the item, algorithm exchange (swap) it with its predecessor.  This results in accumulation of often searched items in the left part of an array and moving the rarely searched items in the right part of an array.  The algorithm, in the self-adaptive manner rebuilds the table according to the probability of keys being searched.
=====================================================================


